# Maintainer: Levente Polyak <anthraxx[at]archlinux[dot]org>
# Maintainer: David Runge <dvzrv@archlinux.org>
# Contributor: Ionut Biru <ibiru@archlinux.org>
# Contributor: Gerardo Exequiel Pozzi <vmlinuz386@yahoo.com.ar>

pkgbase=libcap-ng
pkgname=(libcap-ng python-capng)
pkgver=0.8.3
pkgrel=2
pkgdesc='A library for Linux that makes using posix capabilities easy'
arch=(x86_64)
url='https://people.redhat.com/sgrubb/libcap-ng/'
license=(LGPL2.1)
depends=(glibc)
makedepends=(python swig)
source=(https://github.com/stevegrubb/libcap-ng/archive/v${pkgver}/${pkgname}-${pkgver}.tar.gz)
sha512sums=('5b078164a9e7475349f3143e1f3ee219df09252185726aec92f1e20386e14d1c708f54aea986600078bd5851f6103e43624f8a3527140ebb955601820e2e4c89')
b2sums=('166d4addeea96db2ed9e0f1cb1284af5a2dd690a46ae5fd598e16ea2058adbf94ba523ba594e006eea2a0a6a2f3232f8fb2d4716e749ac3383012af7b0034522')

_pick() {
  local p="$1" f d; shift
  for f; do
    d="$srcdir/$p/${f#$pkgdir/}"
    mkdir -p "$(dirname "$d")"
    mv "$f" "$d"
    rmdir -p --ignore-fail-on-non-empty "$(dirname "$f")"
  done
}

prepare() {
  cd $pkgbase-$pkgver
  autoreconf -fiv
}

build() {
  cd $pkgbase-$pkgver
  ./configure \
    --prefix=/usr \
    --enable-static=no \
    --without-python \
    --with-python3
  make
}

check() {
  make check -C $pkgbase-$pkgver
}

package_libcap-ng() {
  provides=(libcap-ng.so libdrop_ambient.so)

  make DESTDIR="$pkgdir" install -C $pkgbase-$pkgver

  (
    cd "$pkgdir"
    _pick python-capng usr/lib/python*
  )
}

package_python-capng() {
  pkgdesc+=' (Python bindings)'
  depends+=(libcap-ng.so)
  provides=(python-libcap-ng)

  mv -v python-capng/* "$pkgdir"
}

# vim: ts=2 sw=2 et:
